DEPARTMENT OF TRANSPORTATION

COAST GUARD

THE DETERMINATION OF AVERAGE GEOSTROPHIC CURRENT VELOCITIES FROM TEMPORALLY AND SPATIALLY RANDOM HYDROGRAPHIC DATA WITH AN APPLICATION TO THE SOUTHERN CALIFORNIA BIGHT

U.S. Coast Guard
Oceanographic Unit
Washington, D.C.

Oceanographic Unit Technical Report 82-1

THE DETERMINATION OF AVERAGE GEOSTROPHIC CURRENT VELOCITIES FROM TEMPORALLY AND SPATIALLY RANDOM HYDROGRAPHIC DATA WITH AN APPLICATION TO THE SOUTHERN CALIFORNIA BIGHT

BY

DR. KENNETH A. MOONEY

U.S. COAST GUARD OCEANOGRAPHIC UNIT
APPLIED SCIENCE DIVISION

NAVY YARD ANNEX
WASHINGTON, D.C. 20593

U.S. COAST GUARD TECHNICAL REPORT

ABSTRACT

A method employing two-dimensional spline fits of spatially and temporally random hydrographic data is developed in order to be able to determine seasonally averaged geostrophic currents. The method is used in an analysis of the currents in the Southern California Bight.

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

THE DETERMINATION OF AVERAGE GEOSTROPHIC CURRENT VELOCITIES FROM TEMPORALLY AND SPATIALLY RANDOM HYDROGRAPHIC DATA WITH AN APPLICATION TO THE SOUTHERN CALIFORNIA BIGHT

INTRODUCTION

Dynamic depth anomaly data from oceanographic stations are available as a standard product from the National Oceanographic Data Center (NODC). NODC calculates these data from temperature and salinity with depth data which it obtains from various sources. It is highly desirous to analyze these data in a manner such as to obtain seasonally averaged dynamic depth anomaly fields, from which can be determined seasonally averaged geostrophic current velocity fields. The location of the data is, however, generally temporally and spatially random, complicating the averaging procedure, and thus requiring the utilization of special methods.

THE METHOD

The most direct method to analyze the data is to group together all the dynamic depth anomaly data for a particular season and then fit this data by the method of least squares with a two dimensional polynomial of the form:

$$d_D = \sum_{m=o}^{M} \sum_{n=o}^{N} A_{mn} \theta^m \phi^n \qquad (1)$$

where

d_D = dynamic depth anomaly at depth D

A_{mn} = regression coefficients

θ = latitude

ϕ = longitude

M,N = order of the polynomials

The virtue of this method is that it treats each data point for exactly what it is, an independent measure of the dynamic depth field, and at the same time yields an analytic expression for the dynamic depth anomaly field. The latitude and longitude are used as the independent variables because this is the most convenient form in which to obtain the data from NODC and thus no coordinate transformations are necessary. The north-south and east-west geostrophic velocities can now be obtained by differentiating equation (1):

$$U_\theta = \frac{10}{f R \sin\theta} \frac{\partial d_D}{\partial \phi} \qquad (2)$$

$$V_\phi = -\frac{10}{f R} \frac{\partial d_D}{\partial \theta}$$

where

U_θ = north-south velocity

V_ϕ = east-west velocity

f = Coriolis Parameter

R = Radius of the earth

The dynamic depth anomaly field frequently, however, displays so much structure that to replicate it accurately with a single function of the form given in equation (1) would require a polynomial of prohibitively high order. This difficulty can be overcome through the use of two-dimensional spline fits of the data. In this case the area of interest is divided into rectangles and the data in each rectangle is still fit by the method of least squares with a polynomial of the form expressed in equation (1), but with the constraints that the different polynomials and their first and, if necessary, higher derivative match on the boundaries of the rectangles. The order of the polynomial is selected to yield whatever goodness of fit is desired. The advantage of spline fitting is that the data which was previously fitted by one polynomial is now fitted by several polynomials, allowing much more structure to be represented. Equations (2) can still be used to evaluate the components of geostrophic velocity.

THE PROGRAM

To put this method into use a spline fit program called SPLPAK which employs third order polynomials was obtained form the National Center for Atmospheric Research (NCAR) Software Support Library. The SPLPAK program has several options, the most important of which is the number of rectangles into which the area to be fitted is to be divided. Since it was assumed that, even with spline fits, the area to be analyzed would be too large to be fitted by a single spline fit, it was specified for the program to have simply nine rectangles, as shown in Figure 1, with the size of the rectangles being left as a variable to be determined by the needs of the particular area under study. The idea behind using nine rectangles was that the best fit to the data should occur in the center rectangle where the effects of the outer boundary are least significant. The total area of interest would then be analyzed by chosing an appropriate size for the nine rectangles according to the amount of structure present and then moving the center rectangle around the total area of interest with some continuity being supplied from one position of the center rectangle to an adjacent position by means of the overlapping effect of the eight surrounding rectangles. This approach would allow different size areas to be analyzed with equal accuracy and also allow for the development of a generalized program applicable to any area. A fuller explanation of SPLPAK and its other options can be found in Adams et al. (1975).

In order to utilize SPLPAK a program called XTRACT was written to read a standard NODC data tape and write on disk the appropriate input data required by SPLPAK. A program called DYFIT was then written which would read the data from the desk, call SPLPAK as a subroutine and evaluate the components of the geostrophic velocity at any specified location. Besides the size of the rectangles, the relative depth to which the currents are to be referenced and the time period over which the currents are to be averaged are variables which must be specified in DYFIT.

A listing and fuller explanation of the programs XTRACT and DYFIT can be found in Appendices A and B.

THE SOUTHERN CALIFORNIA BIGHT

A region from 32° N to 35° N and from 117°W to 122°W off the coast of Southern California was chosen as the area to test the applicability of the above described method. There were several reasons for choosing this particular area:

- o There have been many hydrographic surveys in the area, so there should be a sufficient amount of data;
- o Qualitatively the currents are well understood showing a definite seasonal dependence;
- o The water is deep so that the geostrophic currents should be meaningful.

A standard hydrographic data tape for the above area was obtained from NODC. After initial preparation of the data with the XTRACT program the DYFIT program was run with rectangles equal to 1° of latitude by 1° of longitude. Also as most of the hydrographic data extended only to 500 meters, 500 meters was chosen as the reference level, thus enabling the utilization of most of the data. While 500 meters may appear to be somewhat shallow for a reference level, cross-checking wherever possible showed no significant differences in surface velocities when 1000 meters was used as a reference level. The spline fit was evaluated at the center of rectangles 1/4° x 1/4°, 1/2° x 1/2° and 1° x 1° on their sides. As it turned out the current structure was such that there were no areas that required rectangles smaller than 1/2° x 1/2° in order to be able to characterize the currents. The data was of sufficient volume to be grouped by months for analysis.

The results of the analysis are depicted in the twelve chartlets in Figures 2-13 with the current directions given in degrees true and the speed in knots. The currents are evaluated at the center of the rectangle. The analysis indicates that the average geostrophic velocities in the Southern California Bight are relatively small, with maximum speeds of only approximately 0.2 knot.

Unfortunately an in-depth critical quantitative comparison of observed and predicted currents is not possible due to a lack of detailed current measurements in the area. The calculated geostrophic velocities do, however, conform with the accepted general circulation scheme for the area as presented for example in Pirie and Stellar (1977). This scheme presents the Southern California Bight as being dominated by a cyclonic gyre, with the southeastwardly flowing California Current to its west, the northwestwardly flowing Davidson Current to its east. The speeds in the gyre are known to be seasonally dependent. On the west the current speeds increase during the spring and early summer as the California Current tends to move eastward during this period. The Davidson Current is an undercurrent during this time period with little surface manifestation. During the fall and winter the Davidson Current surfaces and the California moves westward with a subsequent increase in the current speed on the eastern side of the gyre and a decrease of those on the west. While it is realized that rigorous comparisons are not possible without precise measurements, the above described scheme is well supported qualitatively in the twelve chartlets in Figures 2-13.

CONCLUSIONS

A method has been developed which uses two-dimensional least-squares spline fits of spatially and temporally random dynamic depth anomaly data to determine average geostrophic currents. The method has successfully been applied to quantify the currents in the Southern California Bight. It can be assumed that it is possible to apply this method to other areas and thus begin to utilize the vast amounts of data in the NODC inventory.

REFERENCES

Adams, J. C., A. K. Cline, M. A. Drake and R. A. Sweet, eds., NCAR Software Support Library, NCAR-TN/IA-105, 1975.

Pirie, D. M. and D. D. Stellar, California Coastal Processes Study - Landsat II, Final Report, Landsat Investigation No. 22200, 1977.

APPENDIX A

I. Program Name: XTRACT

II. System Name: DYFIT

III. Purpose: XTRACT is the first step in a system of programs that build a data file for the DYFIT program. XTRACT reads a NOAA STD Data tape and extracts records with valid dynamic depth anomaly data. All other records are skipped. The data is processed (see section VII) and written out onto a disk file for subsequent processing by the IBM SORT routine.

IV. History: XTRACT was originally written in FORTRAN by David A. Portyrata in October 1978. It was rewritten in COBOL in July 1979. The change to COBOL allowed the use of built-in functions to handle the overpunched sign on the latitude and longitude fields, and facilitated use of packed decimal output.

V. Machine: IBM 360/65

VI. Language: IBM COBOL VERSION 4

VII. Description of Processing:

a. Scanning for Valid Records

Many different types of data records are contained on a STD tape. Only record type 6 contains a dynamic depth anomaly field. All other records are skipped. The following conditions are necessary for a record to be processed:

(1) A '6' in the 80th byte (the record type field).
(2) Valid numeric data in bytes 60-63 (the dynamic depth anomaly field).
(3) Valid numeric data in bytes 7-8 (the minutes of latitude field).
(4) Valid numeric data in bytes 13-14 (the minutes of longitude field).

b. Converting Latitude and Longitude Records

Latitude and longitude records are converted from degrees/minutes/tenths of minutes to decimal degrees. The latitudes are subtracted from 90 so that they range from 0 to 180 instead of -90 to 90. 180 is added to the longitudes so that they range from 0 to 360 instead of -180 to 180. The above operations are performed so that the output from XTRACT need not contain signed numeric fields.

c. Computing the Julian Day

The Julian date is calculated using a formula published in the Texas Instruments Master Library for TI 58/59. The factor for the cast day is calculated, and the factor for January 1st of the year the cast was taken is subtracted from it, giving the Julian date. The formula is:

For January and February:

$$FACTOR = 365 (YYYY) + DD + 31 (MM - 1) + INT(YYYY - 1)/4) - INT(3/4[INT] ((YYYY - 1)/100) + 1])$$

For March through December:

$$FACTOR = 365 (YYYY) + DD + 31(MM - 1) - INT(4MM + 23) + INT (YYYY/4) - INT(3/4[INT (YYYY/100) + 1])$$

In these formulas INT indicates using only the integer portion of the expression.

The number of days between two dates is found by calculating the FACTOR for each date, then finding the difference between the FACTORS.

VIII. File Description:

a. Input File

Input to XTRACT is a NOAA tape of STD data. Specific information as to tape characteristics will be on the tape itself. The following characteristics are assumed:

TRACKS	9
DENSITY	1600 Bits/Inch
LOGICAL RECORD LENGTH	80 Bytes
BLOCKSIZE	3200 Bytes
RECORD FORMAT	FB
UNLABELED	

b. Output File

Output from XTRACT is a disk file in packed decimal form. It is to read into the IBM SORT utility. The following characteristics are used:

LOGICAL RECORD LENGTH	18 Bytes
BLOCKSIZE	6372 Bytes

c. Structures Used

The structures used for input and output of data in the EXTRACT program are shown in Tables A1 and A2.

TABLE A1

STD DATA FILE STRUCTURE (XTRIN)
(Logical Record Length = 80 Bytes)

POSITION	DESCRIPTION
01-04	Filler
05-09	Latitude - positions 5 and 6 degrees, positions 7 and 8 minutes, position 9 tenths of minutes, eleven overpunch in position 8 for south latitude.
10-15	Longitude - positions 10-12 degrees, positions 13 and 14 minutes, position 15 tenths of minutes, eleven overpunch in position 14 for east longitude
16-18	Filler
19-24	Date - positions 19 and 20 years, positions 21 and 22 month and positions 23 and 24 day.
25-27	Filler
28-31	Standard Depth
32-59	Filler
60-63	Dynamic depth anomaly
64-79	Filler
80	Record type (6)

TABLE A2

STANDARD DEPTH FILE STRUCTURE (XTROUT)
(Logical Record Length = 18 Bytes)

POSITION	DESCRIPTION
01-03	Standard depth
04&05	Julian day
05&07	Year
08-11	Latitude-decimal degrees
12-15	Longitude-decimal degrees
16-18	Dynamic depth anomaly

```
IDENTIFICATION DIVISION.
PROGRAM-ID. XTRACT.
AUTHOR. BOB STARK, CO-OP, WEST VIRGINIA INSTITUTE OF TECHNOLOGY.
INSTALLATION. US COAST GUARD OCEANOGRAPHIC UNIT, ATD/DPB.
DATE-WRITTEN.  JUL  2, 1980.
DATE-COMPILED.
REMARKS.
        THIS PROGRAM SCANS NODC STD TAPE RECORDS FOR STANDARD
    DEPTH DYNAMIC DEPTH DATA. WHEN A STANDARD DEPTH IS FOUND,
    THE JULIAN DAY FOR THAT RECORD IS CALCULATED. THEN THE
    STANDARD DEPTH, JULIAN DAY, YEAR, LATITUDE, LONGITUDE, AND
    DYNAMIC DEPTH ARE WRITTEN OUT ONTO A FILE IN PACKED DECIMAL
    FORMAT. THIS DATA WILL, AFTER SUBSEQUENT PROCESSING BY CREATE
    AND DUPFIX, BECOME INPUT TO THE DYFIT PROGRAM.
        THIS PROGRAM IS A REWRITE OF THE ORIGINAL FORTRAN XTRACT
    PROGRAM, WRITTEN OCT 2,1979 BY DAVID PORTYRATA. IT WAS
    REWRITTEN IN COBOL TO SIMPLIFY THE HANDLING OF THE OVERPUNCH
    MINUS SIGN IN THE LATITUDE AND LONGITUDE FIELDS.

    * NOTE *...THE JULIAN DAY IS CALCULATED USING AN ALGORITHM
    FROM "T.I. PROGRAMMABLE 58/59 MASTER LIBRARY", PAGE 76.

ENVIRONMENT DIVISION.
CONFIGURATION SECTION.
SOURCE-COMPUTER. IBM-360.
OBJECT-COMPUTER. IBM-360.
INPUT-OUTPUT SECTION.
FILE-CONTROL.
    SELECT STD-DATA-FILE ASSIGN TO UT-S-XTRIN.
    SELECT STANDARD-DEPTH-FILE ASSIGN TO UT-S-XTROUT.

DATA DIVISION.
FILE SECTION.
FD  STD-DATA-FILE LABEL RECORDS OMITTED
    RECORD CONTAINS 80 CHARACTERS
    BLOCK CONTAINS 0 RECORDS.
01  STD-DATA-REC.
    03  FILLER                 PIC X(4).
    03  LATITUDE.
        05  LAT-DEGREES        PIC  9(2).
        05  LAT-MINUTES        PIC S9(2).
        05  LAT-10TH-MIN       PIC  9.
    03  LONGITUDE.
        05  LON-DEGREES        PIC  9(3).
        05  LON-MINUTES        PIC S9(2).
        05  LON-10TH-MIN       PIC  9.
    03  FILLER                 PIC X(3).
    03  DATE-DATA.
        05  DATE-YEAR          PIC 9(2).
        05  DATE-MONTH         PIC 9(2).
        05  DATE-DAY           PIC 9(2).
    03  FILLER                 PIC X(3).
    03  STANDARD-DEPTH         PIC 9(4).
```

```
     03  FILLER               PIC X(28).
     03  DYNAMIC-DEPTH        PIC S9V999.
     03  FILLER               PIC X(16).
     03  STANDARD-DEPTH-FLAG PIC 9.

 FD  STANDARD-DEPTH-FILE LABEL RECORDS STANDARD
     RECORD CONTAINS 18 CHARACTERS
     BLOCK CONTAINS 354 RECORDS.
 01  STANDARD-DEPTH-REC USAGE COMP-3.
     03  ST-STANDARD-DEPTH    PIC 9(5).
     03  ST-JULIAN-DAY        PIC 9(3).
     03  ST-YEAR              PIC 9(2).
     03  ST-LATITUDE          PIC 999V999.
     03  ST-LONGITUDE         PIC 999V999.
     03  ST-DYNAMIC-DEPTH     PIC S99V999.

 WORKING-STORAGE SECTION.
 77  EOF-FLAG                 PIC 9   USAGE COMP SYNC.
 77  MINUTES                  PIC 99  USAGE COMP SYNC.
 77  REC-COUNT                PIC  9(8) VALUE 0 USAGE COMP SYNC.
 77  READ-COUNT               PIC  9(8) VALUE 0 USAGE COMP SYNC.
 77  WS-YEAR                  PIC 9(8)  USAGE COMP SYNC.
 77  WS-LATITUDE              PIC S999V999.
 77  WS-LONGITUDE             PIC S999V999.
 77  REAL-1                             USAGE COMP-1 SYNC.
 77  REAL-2                             USAGE COMP-1 SYNC.
 77  REAL-3                             USAGE COMP-1 SYNC.
 77  INTEGER-1                PIC 9(8)  USAGE COMP SYNC.
 77  INTEGER-2                PIC 9(8)  USAGE COMP SYNC.
 77  INTEGER-3                PIC 9(8)  USAGE COMP SYNC.
 77  JAN-1-FACTOR             PIC S9(8) USAGE COMP SYNC.
 77  FACTOR                   PIC S9(8) USAGE COMP SYNC.

 PROCEDURE DIVISION.
* THIS MODULE IS THE SUPERVISOR. HOUSEKEEPING IS DONE HERE.
     OPEN INPUT STD-DATA-FILE
          OUTPUT STANDARD-DEPTH-FILE.
     PERFORM READ-REC
     PERFORM MAIN-PROCESS THRU MAIN-PROCESS-EXIT
         UNTIL EOF-FLAG = 1.
     DISPLAY 'END OF PROCESSING BY XTRACT ', READ-COUNT,
         ' RECORDS READ ', REC-COUNT, ' RECORDS WRITTEN.'.
     CLOSE STD-DATA-FILE
           STANDARD-DEPTH-FILE.
     STOP RUN.


 MAIN-PROCESS.
* THIS MODULE MAKES SURE THE DATA IS A STANDARD DEPTH.
*    AND CONVERTS THE LATITUDE AND LONGITUDE TO THE PROPER FORM.
     ADD 1 TO READ-COUNT.

 EDIT-CHECK-THE-DATA.
     IF LAT-DEGREES IS NOT NUMERIC
```

```
          OR LON-DEGREES IS NOT NUMERIC
          OR DATE-DATA IS NOT NUMERIC
          OR STANDARD-DEPTH IS NOT NUMERIC
          OR STANDARD-DEPTH-FLAG IS NOT NUMERIC
       THEN PERFORM READ-REC
            GO TO MAIN-PROCESS-EXIT.
*      CHECK FOR A STANDARD DEPTH. IF NOT, READ NEXT REC, AND EXIT.
       IF STANDARD-DEPTH-FLAG NOT = 6
          OR DYNAMIC-DEPTH IS NOT NUMERIC
       THEN PERFORM READ-REC
            GO TO MAIN-PROCESS-EXIT.

*      CONVERT DEGREES-MINUTES-TENTH-MINUTES OF LATITUDE TO DECIMAL
*      DEGREES. (NOTE...MINUTES IS AN UNSIGNED FIELD.)
       IF LAT-MINUTES IS NOT NUMERIC
       THEN PERFORM READ-REC
            GO TO MAIN-PROCESS-EXIT
       ELSE MOVE LAT-MINUTES TO MINUTES.
*       CONVERT BLANKS TO ZERO.
       IF LAT-10TH-MIN IS NOT NUMERIC
       THEN MOVE 0 TO LAT-10TH-MIN.
       COMPUTE WS-LATITUDE = LAT-DEGREES + (MINUTES / 60)
           + (LAT-10TH-MIN / 600).
       IF LAT-MINUTES IS NEGATIVE
       THEN MULTIPLY WS-LATITUDE BY -1 GIVING WS-LATITUDE.
*      90 IS ADDED TO ALL LATITUDES SO THEY WILL BE POSITIVE
*      AND RANGE FROM 0 TO 180.
       ADD 90, WS-LATITUDE GIVING ST-LATITUDE.

*      CONVERT DEGREES-MINUTES-TENTHS OF LONGITUDE TO DECIMAL DEGREE
       IF LON-MINUTES IS NOT NUMERIC
       THEN PERFORM READ-REC
            GO TO MAIN-PROCESS-EXIT
       ELSE MOVE LON-MINUTES TO MINUTES.
*       CONVERT BLANKS TO ZERO.
       IF LON-10TH-MIN IS NOT NUMERIC
       THEN MOVE 0 TO LON-10TH-MIN.
       COMPUTE WS-LONGITUDE = LON-DEGREES + (MINUTES / 60)
           + (LON-10TH-MIN / 600).
       IF LON-MINUTES IS NOT NEGATIVE
       THEN MULTIPLY WS-LONGITUDE BY -1 GIVING WS-LONGITUDE.
*      180 IS ADDED TO ALL LONGITUDES SO THAT THEY WILL
*      BE POSITIVE AND RANGE FROM 0 TO 360.
       ADD 180, WS-LONGITUDE GIVING ST-LONGITUDE.

   COMPUTE-JULIAN-DAY.
*          FIRST WE FIND THE JULIAN DATE FACTOR FOR JANUARY 1ST OF
*      THE YEAR THE CAST WAS TAKEN.
       ADD 1900, DATE-YEAR GIVING WS-YEAR.
       COMPUTE INTEGER-1 = (WS-YEAR - 1) / 4.
       COMPUTE INTEGER-2 = ((WS-YEAR - 1) / 100) + 1
       COMPUTE REAL-2 = INTEGER-2 * 0.75.
       MOVE REAL-2 TO INTEGER-2.
       COMPUTE JAN-1-FACTOR = (365 * WS-YEAR) + INTEGER-1 + 1 -
```

```
-              INTEGER-2.

*          NOW WE FIND THE JULIAN DATE FACTOR FOR THE DAY THE CAST
*     WAS TAKEN.
      IF DATE-MONTH < 3
      THEN PERFORM JAN-FEB-FACTOR
      ELSE PERFORM MAR-DEC-FACTOR.

*     NOW WE FIND THE DIFFERENCE IN THE FACTORS, THE JULIAN DAY
      COMPUTE ST-JULIAN-DAY = FACTOR - JAN-1-FACTOR + 1.

*     MOVE THE REST OF THE FIELDS TO THE OUTPUT BUFFER.
      MOVE DATE-YEAR TO ST-YEAR.
      MOVE STANDARD-DEPTH TO ST-STANDARD-DEPTH.
      MOVE DYNAMIC-DEPTH TO ST-DYNAMIC-DEPTH.

*     WRITE OUT THE RECORD & READ IN DATA FOR THE NEXT LOOP.
      WRITE STANDARD-DEPTH-REC.
      ADD 1 TO REC-COUNT.
      PERFORM READ-REC.

 MAIN-PROCESS-EXIT.
      EXIT.

 JAN-FEB-FACTOR.
*         THIS MODULE COMPUTES JULIAN DATE FACTORS FOR JANUARY AND
*     FEBRUARY. ALGORITHM FROM TEXAS INSTRUMENTS MASTER LIBRARY.
*     NOTE...INTEGER-1 & 2 WERE ALREADY CALCULATED WHEN
*     JAN-1-FACTOR WAS CALCULATED.
      COMPUTE FACTOR = (365 * WS-YEAR) + (31 * (DATE-MONTH - 1)
-     ) + DATE-DAY + INTEGER-1 - INTEGER-2.

 MAR-DEC-FACTOR.
*         THIS MODULE COMPUTES JULIAN DATE FACTORS FOR MARCH THRU
*     DECEMBER. ALGORITHM FROM TEXAS INSTRUMENTS MASTER LIBRARY.
      COMPUTE REAL-1 = (0.4 * DATE-MONTH) + 2.3.
      MOVE REAL-1 TO INTEGER-1.
      COMPUTE INTEGER-2 = WS-YEAR / 4.
      COMPUTE INTEGER-3 = (WS-YEAR / 100) + 1.
      COMPUTE REAL-3 = INTEGER-3 * 0.75.
      MOVE REAL-3 TO INTEGER-3.
      COMPUTE FACTOR = (365 * WS-YEAR) + (31 * (DATE-MONTH - 1)
-     ) + DATE-DAY - INTEGER-1 + INTEGER-2 - INTEGER-3.

 READ-REC.
*     THIS MODULE READS RECORDS FROM THE STD DATA TAPE.
      READ STD-DATA-FILE AT END MOVE 1 TO EOF-FLAG.
```

I. Program Name: DYFIT

II. System Name: DYFIT

III. Purpose: The DYFIT program calculates the average geostrophic velocity field from randomly located dynamic depth anomaly data.

IV. History: The DYFIT program was written by K. A. Mooney for the CDC 3300 computer in the summer of 1978. Due to lack of memory and slow processing speed of the CDC, this version was impractical for production use. The IBM 360 version was written by K. A. Mooney in the spring of 1979. This version was used to study currents in the Southern California, Gulf of Mexico, and Hawaii regions. This version searched a large sequential file (created by XTRACT or XTEMP) for suitable dynamic depth anomaly data. This program was highly I/O bound because it only used about 0.5% of the records that it has to read. The current version of DYFIT was written by Bob Stark in the summer of 1980. This version utilizes a COBOL subroutine which reads small sections of an indexed sequential file. This reduced the I/O time required to run the program. The XTRACT and XTEMP programs were rewritten and the CREATE and DUPFIX programs written to build indexed sequential files for this version.

V. Machine: IBM 360/65

VI. Language:

a. Languages Used

1. IBM 'H' level FORTRAN (Optimized)
2. IBM version 4 ANSI COBOL (Optimized)

b. Details of FORTRAN-COBOL linkage

1. Compilation

The COBOL modules (DYFIT and SETKEY/READER) are compiled in separate runs of the COBOL compiler. The object module produced by the first run (ddname = SYSLIN) is passed on to the second run. All of the FORTRAN subroutines are compiled in a single run of the FORTRAN compiler and placed in the same data set as the COBOL object modules (SYSLIN). The linkage editor is then run to link all of the object modules into a load module and store it on disk. The details of this process may be seen in the cataloged procedure K198PDFT, which is used to put the DYFIT program to disk.

Transfer of control from main program to subroutine and back is accomplished by the DYFIT program, which calls a ILBOSTPO (a library subroutine) to initialize the subroutine communications save area. This assures that all routines called by DYFIT, or called by routines called by DYFIT, will be treated as subroutines. This is especially important for the SETKEY/READER routine, as EXIT PROGRAM statements in the subroutine communications save area in order for control to return to the calling routine.

3. Correspondence of Identifiers

Two types of identifiers were used as parameters in the COBOL routines to be called by FORTRAN routines: Fullword real (floating point) and halfword integer (binary). These variables must be aligned on fullword boundaries, which is automatic in FORTRAN. In COBOL the keyword SYNCRONIZED (abbreviated SYNC) is used to align the identifiers on a fullword boundary.

The fullword real identifiers are declared as REAL*4 in the FORTRAN programs, and defined as USAGE COMP-1 SYNC in the COBOL routines. These two definitions are fully compatable.

The halfword integer identifiers are declared as INTEGER*2 in the FORTRAN programs, and defined as PIC S9999 USAGE COMP SYNC in the COBOL routines.

VII. Description of Processing: DYFIT calls DYFORT which reads a parameter card specifying a latitude, longitude, and time range. This card also indicates from what standard depth the data is to come. SETKEY is called to position the indexed sequential file at the first record of the user specified standard depth that falls within the time range chosen. The file is searched from this point for dynamic depth anomaly data that falls within the specified latitude and longitude. The search stops when the given time range is exceeded, the given standard depth is exceeded, or the physical end of file is reached. SPLCW is then called to fit a two dimensional cubic spline to the dynamic depth anomaly data. A description of SPLCW can be found in the chapter on SPLPAK in volume 1 of the NCAR SOFTWARE SUPPORT LIBRARY, NCAR technical note IA-105, March, 1975. Finally, DYFORT evaluates the first derivatives of the spline fit at user specified locations by calling the function SPLDE. A description of SPLDE can also be found in the above reference. The components of the geostrophic velocities can then be determined from the formulae:

$$U_\theta = \frac{10}{f R \sin\theta} \frac{\partial d_D}{\partial \phi}$$

$$V_\phi = - \frac{10}{f R} \frac{\partial d_D}{\partial \theta}$$

where

U_θ = North - South component of geostrophic velocity

V_ϕ = East - west component of geostrophic velocity

f = Coriolis parameter

R = Radius of earth

d_D = Dynamic depth anomaly field at depth D

ϕ = Longitude

θ = Latitude

The printout consists of the standard depth of the data and the center and ranges for the latitude, longitude, and Julian day. At the end of the data list (appears if data print option was selected), the latitude, longitude, and Julian day at which the plane fit was evaluated along with the components of geostrophic velocity in knots and the number of points which were used in the fit are printed.

VIII. File Description:

a. Disk File

1. Physical Characteristics

(a) Indexed sequential organization
(b) Sequential access mode
(c) One cylinder used for cylinder index
(d) RECFM = FB (fixed length blocks)
(e) LRECL = 18 bytes (logical record length)
(f) KEYLENGTH = 15 bytes
(g) BLKSIZE = 6372 bytes
(h) Blocking factor 354 records/block
(i) All fields in packed decimal form.

2. Usage

The dynamic-height-file (ISAMFILE) contains the dynamic depth anomaly data for DYFIT. It is created by the CREATE program - refer to it for sort order. In DYFIT, a call to SETKEY opens the file and positions it at the first record matching the starting - standard - depth (DPDP) and the first - Julian - day (TMIN) parameters. If there is no data in the file at this position, SETKEY returns a 1 in key - error (KEYERR). READER is then called to read records sequentially from that point on until a record is read whose Julian day is greater than last - Julian - day (TMAX), whose standard - depth is greater than starting - STD - depth (DPDP), or the end of file is found. The file is then closed and reader returns a 1 in EOF.

b. Card File

1. Physical Characteristics

(a) Sequential organization
(b) LRECL = 80 bytes (logical record length)
(c) Unblocked

c. Structures Used

1. The structures used for the input files are shown in Tables B1, B2 and B3.

TABLE B1

DYNAMIC HEIGHT FILE (ISAMFILE)

POSITION	DESCRIPTION
01-03	Standard depth
04&05	Julian day
06&07	Year
08-11	Latitude
12-15	Longitude
16-18	Dynamic depth

TABLE B2

PARAMETER CARD (SYSIN)

POSITION	DESCRIPTION
01-05	Standard depth
06-08	Filler
09-14	Center latitude
15	Filler
16-22	Center longitude
23	Filler
24-27	Center Julian day
28	Filler
29-33	Latitude range
34	Filler
35-40	Longitude range
41	Filler
42-45	Julian day range
46	Filler
47	Print
48-80	Filler

TABLE B3

DATA CARD (SYSIN)

POSITION	DESCRIPTION
01-06	Latitude
7	Filler
08-14	Longitude
15-80	Filler

```
IDENTIFICATION DIVISION.
PROGRAM-ID. DYFIT.
AUTHOR. BOB STARK, CO-OP, WEST VIRGINIA INSTITUTE OF TECHNOLOGY.
INSTALLATION. US COAST GUARD OCEANOGRAPHIC UNIT. ATD/DPB.
DATE-WRITTEN.  JUN 26,1980.
DATE-COMPILED.
REMARKS.  THIS PROGRAM IS THE MAIN PROGRAM OF THE DYFIT SYSTEM.
    IT HAS 2 STATEMENTS - A CALL TO DYFORT (THE OLD MAIN ROUTINE)
    AND A STOP STATEMENT.
        THE FIRST COBOL PROGRAM IN A MIXED LANGUAGE ENVIRONMENT
    CALLS ILBOSTPO (A LIBRARY SUBROUTINE) WHICH INITIALIZES THE
    SUBROUTINE COMMUNICATIONS AREA AND SAVES POINTER REGISTERS
    NECESSARY FOR RETURN TO THE MAIN PROGRAM. THEREFORE THE FIRST
    PROGRAM MUST BE A COBOL PROGRAM, AS CONTROL WOULD NOT RETURN
    TO A MAIN FORTRAN PROGRAM AFTER IT HAD CALLED A COBOL SUB-
    ROUTINE. FOR MORE INFO. REFER TO CALLING AND CALLED PROGRAMS.
    IBM OS ANSI COBOL PROGRAMMERS GUIDE.

ENVIRONMENT DIVISION.
DATA DIVISION.
PROCEDURE DIVISION.
    CALL 'DYFORT'.
    STOP RUN.


IDENTIFICATION DIVISION.
PROGRAM-ID. SETKEY.
AUTHOR. CO-OP BOB STARK USCG OCEANOGRAPHIC UNIT.
INSTALLATION. US COAST GUARD OCEANOGRAPHIC UNIT. ATD/DPB.
DATE-WRITTEN. 6/11/1980.
DATE-COMPILED.
REMARKS.      THIS PROGRAM READS AN INDEXED SEQUENTIAL FILE
    CONTAINING LATITUDE, LONGITUDE, YEAR, JULIAN-DAY, STANDARD
    DEPTH, AND DYNAMIC HEIGHT RECORDS;  THE FILE IS SORTED BY
    STANDARD DEPTH, JULIAN DAY, LATITUDE, AND LONGITUDE; SKIP
    SEQUENTIAL LOGIC IS USED TO POSITION THE FILE AT THE PROPER
    START POINT;
    TWO ENTRY POINTS ARE USED:
        1)  SETKEY - POSITIONS THE FILE AT THE FIRST DESIRED
                     RECORD BASED ON THE STD.DEPTH AND JULIAN DAY
        2)  READER - READS THE RECORDS AND RETURNS AN END OF
                     FILE FLAG;  CLOSES THE FILE WHEN EOF=1.
ENVIRONMENT DIVISION.
CONFIGURATION SECTION.
SOURCE-COMPUTER. IBM-360.
OBJECT-COMPUTER. IBM-360.
INPUT-OUTPUT SECTION.
FILE-CONTROL.
    SELECT DYNAMIC-HEIGHT-FILE ASSIGN TO DA-I-ISAMFILE
        ACCESS MODE IS SEQUENTIAL
        RECORD KEY IS RECORD-KEY.

DATA DIVISION.
```

```
 FILE SECTION.
 FD  DYNAMIC-HEIGHT-FILE LABEL RECORDS STANDARD
     RECORD CONTAINS 18 CHARACTERS
     BLOCK CONTAINS 354 RECORDS.
 01  DYNAMIC-HEIGHT-REC.
     03  RECORD-KEY.
         05  STANDARD-DEPTH  PIC 9(5)        USAGE COMP-3 SYNC.
         05  JULIAN-DAY      PIC 9(3)        USAGE COMP-3 SYNC.
         05  YEAR            PIC 9(2)        USAGE COMP-3.
         05  LATITUDE        PIC  999V999    USAGE COMP-3 SYNC.
         05  LONGITUDE       PIC  999V999    USAGE COMP-3 SYNC.
     03  DYNAMIC-HEIGHT      PIC S99V999     USAGE COMP-3.

 WORKING-STORAGE SECTION.
 01  EOF-FLAG                PIC S9999       USAGE COMP   SYNC.
 01  GENERIC-KEY.
     03  NOM-STD-DEPTH           PIC 9(5)    USAGE COMP-3 SYNC.
     03  NOM-FIRST-JULIAN-DAY    PIC 9(3)    USAGE COMP-3 SYNC.

 LINKAGE SECTION.
*    THESE ARE USED WHEN SETKEY IS CALLED.
 01  FIRST-JULIAN-DAY                        USAGE COMP-1 SYNC.
 01  LAST-JULIAN-DAY                         USAGE COMP-1 SYNC.
 01  STARTING-STD-DEPTH                      USAGE COMP-1 SYNC.
 01  KEY-ERROR               PIC S9999       USAGE COMP   SYNC.
*    THESE ARE USED WHEN READER IS CALLED.
 01  LINK-LATITUDE                           USAGE COMP-1 SYNC.
 01  LINK-LONGITUDE                          USAGE COMP-1 SYNC.
 01  LINK-DYNAMIC-HEIGHT                     USAGE COMP-1 SYNC.
 01  LINK-EOF-FLAG           PIC S9999       USAGE COMP   SYNC.


*SETKEY ROUTINE
 PROCEDURE DIVISION USING FIRST-JULIAN-DAY, LAST-JULIAN-DAY,
                          STARTING-STD-DEPTH, KEY-ERROR.
     MOVE ZERO TO EOF-FLAG, KEY-ERROR.
     OPEN INPUT DYNAMIC-HEIGHT-FILE.
     MOVE STARTING-STD-DEPTH TO NOM-STD-DEPTH.
     MOVE FIRST-JULIAN-DAY TO NOM-FIRST-JULIAN-DAY.
     START DYNAMIC-HEIGHT-FILE USING KEY RECORD-KEY = GENERIC-KEY
           INVALID KEY MOVE 1 TO KEY-ERROR
                    CLOSE DYNAMIC-HEIGHT-FILE.
     GO TO SETKEY-EXIT.

 SETKEY-EXIT.
     EXIT PROGRAM.


 READER-ROUTINE.
     ENTRY 'READER' USING LINK-LATITUDE, LINK-LONGITUDE,
                    LINK-DYNAMIC-HEIGHT, LINK-EOF-FLAG.
     MOVE EOF-FLAG TO LINK-EOF-FLAG.
```

```
      READ DYNAMIC-HEIGHT-FILE
           AT END PERFORM CLOSE-UP
                  GO TO EXIT-READER.

      IF STANDARD-DEPTH > STARTING-STD-DEPTH
      THEN PERFORM CLOSE-UP
           GO TO EXIT-READER
      ELSE IF JULIAN-DAY > LAST-JULIAN-DAY
           THEN PERFORM CLOSE-UP

*                SUBTRACT 90 FROM LATITUDE, 180 FROM LONGITUDE TO
*                CHANGE THE RANGE FROM 0-180 AND 0-360 TO -90 TO +90
*                AND -180 TO +180.
           ELSE SUBTRACT 90 FROM LATITUDE GIVING LINK-LATITUDE
                SUBTRACT 180 FROM LONGITUDE GIVING LINK-LONGITUDE
                MOVE DYNAMIC-HEIGHT TO LINK-DYNAMIC-HEIGHT.

  EXIT-READER.
      EXIT PROGRAM.

  CLOSE-UP.
      MOVE 1 TO EOF-FLAG, LINK-EOF-FLAG.
      CLOSE DYNAMIC-HEIGHT-FILE.
```

```
C***********************************************************************
C*                                                                     *
C*                          PROGRAM DYFIT                              *
C*                                                                     *
C*                       WRITTEN BY KEN MOONEY                         *
C*                                                                     *
C*  CHANGES                                                            *
C*  1)   CHANGED FROM MAIN ROUTINE TO SUBROUTINE TO BE CALLED BY A     *
C*       MAIN COBOL ROUTINE IN ORDER TO FACILITATE FORTRAN AND COBOL   *
C*       SUBROUTINE COMMUNICATION.                                     *
C*  2)   ALL CALLS TO SPL4 HAVE BEEN DELETED AS SPL4  HAS BEEN         *
C*       DELETED FROM THE DYFIT SYSTEM.                                *
C*                  ENTERED 6/10/1980   BOB STARK                      *
C*  3)   ADDED STANDARD DEPTH OF ZERO METERS. ENTERED 7/1/1980 B. STARK
C***********************************************************************
      SUBROUTINE DYFORT
      COMMON A(169)
      COMMON XDATA(3,1000),YDATA(1000),WDATA(1000)
      COMMON WORK(16)
      DIMENSION Z(18)
      DATA Z/0.,20.,30.,50.,75.,100.,125.,150.,200.,250.,300.,400.,500.,
     A600.,700.,800.,900.,1000./
C*       DPDP      STANDARD DEPTH
C*       ALAT      CENTER LATITUDE
C*       ALON      CENTER LONGITUDE
C*       CTIME     CENTER IN TIME (JULIAN DAY)
C*       YRNG      RANGE FOR LATITUDE
C*       XRNG      RANGE FOR LONGITUDE
C*       TRNG      RANGE FOR TIME
C*       IPRT      IF 1 DATA PULLED OFF FILE BY SPL3 WILL BE PRINTED
 10   READ(1,900,END=40)DPDP,ALAT,ALON,CTIME,YRNG,XRNG,TRNG,IPRT
  900 FORMAT(F5.0,2X,   1X,F6.2,1X,F7.2,1X,F4.0,1X,F5.2,1X,F6.2,1X,
     1F4.0,1X,I1)
      IF(DPDP.EQ.9999.) GO TO 40
      PRINT 903,DPDP,ALAT,YRNG,ALON,XRNG,CTIME,TRNG
 903  FORMAT('1 STANDARD DEPTH = ',F5.0,7X,'LATITUDE = ',F6.2,' +/- ',
     A F5.2,7X,'LONGITUDE = ',F7.2,' +/- ',F6.2,6X,'JULIAN DAY = ',
     B F4.0,' +/- ',F4.0)
C        THIS SECTION CHECKS TO MAKE SURE  THAT DPDP IS A STANDARD DEPTH
      DO 20 I=1,18
      IF(DPDP.EQ.Z(I)) GO TO 30
   20 CONTINUE
      PRINT 902
 902  FORMAT(' DEPTH ENTERED IS NOT A STANDARD DEPTH.')
C        DPDP WASN'T A STANDARD DEPTH. NOW WE READ AND DUMP RECORDS
C      UNTIL WE FIND A 99 CARD WHICH MARKS THE END OF THIS SET OF POINTS
 25   READ(1,901,END=40) DPDP
 901  FORMAT(F6.2)
      IF (DPDP.EQ.99.) GO TO 10
      GO TO 25
C
   30 CALL SPL3(DPDP,ALAT,ALON,CTIME,XRNG,YRNG,TRNG,IPRT)
      GO TO 10
```

```
C
 40   RETURN
      END
      SUBROUTINE SPL3(DPDP,ALAT,ALON,CTIME,XRNG,YRNG,TRNG,IPRT)
CC*****************************************************************************
      COMMON A(169)
      COMMON XDATA(3,1000),YDATA(1000),WDATA(1000)
      COMMON WORK(16)
      DIMENSION XMIN(2),XMAX(2),NODES(2),NDERIV(2),COEF(16),X(2)
      DATA NDIM/2/,NODES(1)/4/,NODES(2)/4/,XTRAP/1./,NCF/16/,NWRK/17/
      REAL MINLAT, MINLON, MAXLAT, MAXLON
      INTEGER*2 EOF, KEYERR
CC*****************************************************************************
C   THIS SECTION REWRITTEN 6-13-1980 BY COOP BOB STARK TO PROVIDE
C   COMPATABILITY WITH NEW COBOL INDEXED SEQUENTIAL RECORD RETRIEVAL
C   SUBPROGRAMS ADDED TO THE DYFIT SYSTEM.
      XMIN(2)=ALAT-YRNG
      MINLAT = XMIN(2)
      XMAX(2)=ALAT+YRNG
      MAXLAT = XMAX(2)
      XMIN(1)=ALON-XRNG
      MINLON = XMIN(1)
      XMAX(1)=ALON+XRNG
      MAXLON = XMAX(1)
      TMIN=CTIME-TRNG
      TMAX=CTIME+TRNG
      TMIN2 = 0.
      TMAX2 = 0.
      NDATA=0
C   CHECK FOR JANUARY TO DECEMBER WRAPAROUND IN START DATE.
      IF(TMIN.GE.1) GO TO 8
             TMIN=TMIN+365.
             TMIN2=1.
             TMAX2=TMAX
             TMAX=365.
             GO TO 10
C   CHECK FOR DECEMBER TO JANUARY WRAPAROUND IN THE END DATE.
 8    IF (TMAX.LE.365.) GO TO 10
             TMIN2 = 1.
             TMAX2 = TMAX-365.
             TMAX = 365.
C
C   SET THE STARTING FILE KEY TO DESIRED DYNAMIC DEPTH
C   AND JULIAN DAY (TMIN).
 10   CALL SETKEY(TMIN, TMAX, DPDP, KEYERR)
      IF (KEYERR.NE.1) GO TO 20
C         IF WE GOT HERE, THE SETKEY ROUTINE COULD NOT FIND A KEY AT
C   THE STARTING KEY POINT. NOW WE TRY INCREMENTING THE START DAY WITH
C   THE HOPE OF FINDING A GOOD KEY. IF WE CAN'T FIND A GOOD KEY FROM
C   TMIN TO TMAX, THE AREA MUST BE OUTSIDE THE RANGE OF THE FILE.
      TMIN = TMIN+1.
      IF (TMIN.LT.TMAX) GO TO 10
C         O. K. WE BOMBED OUT ON THAT ATTEMPT TO SET THE KEY.
C   CHECK FOR A 2 ENTRY POINT CONDITION (IE DECEMBER-JANUARY WRAPAROUND)
```

```
C    IF FOUND, BRANCH TO THE SECOND KEY SETUP ROUTINE
        IF (TMIN2.NE.0..OR.TMAX2.NE.0.) GO TO 22
C
C    SORRY - CAN'T SET A FILE KEY.
        PRINT 15,DPDP
 15     FORMAT (' NO DATA IN FILE FOR STD DEPTH = ',F6.0)
C    READ AND DUMP POINTS UNTIL A 99 IS FOUND.
 17     READ 930,X(2)
        IF(X(2).NE.99.) GO TO 17
        GO TO 70
C
C
C    THIS IS THE MAIN READ LOOP.
 20     CALL READER (ALATT, ALONG, DYNHGT, EOF)
        IF (EOF.NE.1) GO TO 25
C          FILE IS AT END
C    CHECK FOR 2 START POINTS.
        IF   (TMIN2.EQ.0..AND.TMAX2.EQ.0.) GO TO 40
C          FILE DOES HAVE 2  START POINTS DUE TO DECEMBER-JANUARY
C    WRAPAROUND. NOW WE HAVE TO SET THE KEY FOR THE SECOND
C    START POINT OF THE FILE.
 22     TMIN = TMIN2
        TMAX = TMAX2
        TMIN2=0.
        TMAX2 = 0.
        GO TO 10
C
C          THIS SECTION CHECKS THE DATA FOR VALIDITY.
C    THE NEXT STATEMENT ADJUSTS THE LONGITUDE IF THE CENTER
C    IS  EAST OF THE INTERNATIONAL DATE LINE AND THE POINT IS WEST.
 25     IF (MINLON.LT.-180..AND.ALONG.GT.0.) ALONG =-360.+ALONG
C    IS    THE NEXT STATEMENT ADJUSTS THE LONGITUDE IF THE CENTER IS
C    WEST OF THE INTERNATIONAL DATE LINE AND THE POINT IS EAST.
        IF (MAXLON.GT.180..AND.ALONG.LT.0.) ALONG = 360.+ALONG
C
C          NOW CHECK THAT LAT AND LONG ARE WITHIN RANGE.
        IF (ALATT.GT.MAXLAT.OR.ALONG.GT.MAXLON.OR.ALATT.LT.MINLAT.OR.
       1 ALONG.LT.MINLON) GO TO 20
C
C    O. K. WE MADE IT. DATA IS GOOD.
        NDATA = NDATA+1
        WDATA(NDATA) = 1.
        XDATA(2,NDATA) = ALATT
        XDATA(1,NDATA) = ALONG
        YDATA(NDATA) = DYNHGT
        IF (NDATA.GE.1000) GO TO 40
        GO TO 20
C
C    RESUME DR. KEN MOONEY'S ORIGINAL SPL3 ROUTINE.
   40 IF(IPRT.NE.1) GO TO 50
        PRINT 910
  910 FORMAT(1H-,10X,3HLAT,8X,4HLONG, 6X,6HDY HGT,//)
        PRINT 920,(I,XDATA(2,I),XDATA(1,I),YDATA(I),I=1,NDATA)
  920 FORMAT(1H ,I3,5X,F6.2,5X,F7.2, 5X,F8.3)
```

```
   50 CALL SPLCW(NDIM,NDATA,XMIN,XMAX,NODES,XTRAP,COEF,NCF,NWRK,IERROR)
      IF(IERROR.GE.32) GO TO 60
      PRINT 940
  940 FORMAT(1H1,2X,3HLAT,8X,4HLONG,7X,3HDAY,6X,3HDIR,7X,3HSPD,7X,
     11HU,9X,1HV,6X,5HNDATA)
C        READ THE LATITUDE AND LONGITUDE AT WHICH THE SURFACE CURRENT
C     IS TO BE COMPUTED.
   60 READ 930,X(2),X(1)
  930 FORMAT(F6.2,1X,F7.2)
      IF(X(2).EQ.99.) GO TO 70
      IF(IERROR.GE.32) GO TO 60
      CON1=-1.2/SIN(X(2)/57.296)
      CON2=2.4/SIN(2.*X(2)/57.296)
      NDERIV(1)=0
      NDERIV(2)=1
      VX=CON1*SPLDE(NDIM,X,NDERIV,COEF,XMIN,XMAX,NODES,IERROR)
      IF(IERROR.GE.32) GO TO 70
      NDERIV(1)=1
      NDERIV(2)=0
      VY=CON2*SPLDE(NDIM,X,NDERIV,COEF,XMIN,XMAX,NODES,IERROR)
      IF(IERROR.GE.32) GO TO 70
      VEL=(VX*VX+VY*VY)**0.5
      ANG=57.296*ATAN2(VX,VY)
      IF(ANG.LT.0.)ANG=ANG+360.
      PRINT 950,X(2),X(1),CTIME,ANG,VEL,VX,VY,NDATA
  950 FORMAT(1H0,F6.2,5X,F7.2,5X,F4.0,5X,F4.0,3(5X,F5.2),5X,I3)
      GO TO 60
   70 RETURN
      END
      SUBROUTINE SPLCW(NDIM,NDATA,XMIN,XMAX,
     1                  NODES,XTRAP,COEF,NCF,NWRK,IERROR)
C DESCRIPTION OF SPLCW IS AVAILABLE IN NCAR SOFTWARE SUPPORT
C LIBRARY, NCAR TECHNICAL NOTE IA-105, MARCH 1975.
```

LIST OF CAPTIONS

Figure 1. Node arrangement used in spline fit analysis.

Figure 2. Average geostrophic currents in the Southern California Bight for the month of January.

Figure 3. Average geostrophic currents in the Southern California Bight for the month of February.

Figure 4. Average geostrophic currents in the Southern California Bight for the month of March.

Figure 5. Average geostrophic currents in the Southern California Bight for the month of April.

Figure 6. Average geostrophic currents in the Southern California Bight for the month of May.

Figure 7. Average geostrophic currents in the Southern California Bight for the month of June.

Figure 8. Average geostrophic currents in the Southern California Bight for the month of July.

Figure 9. Average geostrophic currents in the Southern California Bight for the month of August.

Figure 10. Average geostrophic currents in the Southern California Bight for the month of September.

Figure 11. Average geostrophic currents in the Southern California Bight for the month of October.

Figure 12. Average geostrophic currents in the Southern California Bight for the month of November.

Figure 13. Average geostrophic currents in the Southern California Bight for the month of December.





JANUARY MEAN CURRENT
122°W
121°
120°
119°
118°
117°W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
103
.02
001
.01
342
.02
348
.03
143
.02
063
.01
004
.01
348
.03
013
.03
107
.06
082
.05
094
.04
049
.03
025
.02


FEBRUARY MEAN CURRENT
122°W
121°
120°
119°
118°
117°W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
134
.02
019
.01
356
.01
018
.01
128
.03
123
.02
029
.02
018
.01
039
.01
083
.06
074
.03
111
.08
118
.07
115
.07
136
.05
085
.02
053
.01


MARCH MEAN CURRENT
122°W
121°
120°
119°
118°
117°W
35°
N
34°
33°
32°
C A L I F O R N I A
MEXICO
132 .05
124 .02
074 .02
064 .01
064 .01
037 .01
119 .07
096 .12
131 .03
102 .02
064 .01
051 .01
111 .05
129 .10
106 .07
100 .02
134 .05
141 .07
106 .03
085 .01



APRIL MEAN CURRENT
122° W
121°
120°
119°
118°
117° W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
172 .02
134 .06
134 .06
160 .03
003 .01
035 .01
035 .01
036 .01
130 .08
120 .17
098 .03
081 .03
074 .02
056 .01
135 .08
127 .10
125 .11
123 .11
114 .03
159 .06
171 .06
131 .06
120 .09
102 .05
087 .02


MAY MEAN CURRENT
122° W
121°
120°
119°
118°
117° W
35° N
34°
33°
32°
CALIFORNIA
MEXICO
132 .05
136 .05
141 .01
118 .08
119 .08
109 .06
115 .04
069 .02
022 .02
119 .11
129 .11
136 .03
133 .02
118 .11
113 .13
125 .16
116 .11
126 .06
022 .02
029 .02
120 .10
111 .13
122 .11
118 .14
105 .05
074 .04
106 .08
105 .09
120 .07
125 .08
147 .11
114 .05
095 .04
050 .02


JUNE MEAN CURRENT
122° W
121°
120°
119°
118°
117° W
35° N
34°
33°
32°
CALIFORNIA
MEXICO
154 .05
161 .04
137 .01
126 .08
124 .08
109 .07
114 .04
022 .01
037 .02
113 .10
128 .07
037 .02
019 .02
114 .20
119 .22
129 .13
145 .11
083 .02
034 .02
013 .02
116 .09
118 .12
133 .11
126 .13
125 .04
133 .05
148 .07
147 .08
158 1.0
099 .03
030 .02



JULY MEAN CURRENT
122° W
121°
120°
119°
118°
117° W
35°
N
34°
33°
32°
C A L I F O R N I A
MEXICO
079 .01
139 .06
120 .02
195 .01
347 .03
357 .02
121 .10
156 .06
046 .01
026 .01
114 .15
136 .18
159 .15
123 .03
295 .01
357 .02
011 .02
108 .09
141 .12
122 .09
023 .03
019 .02


AUGUST MEAN CURRENT
122° W
121°
120°
119°
118°
117° W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
123 .04
343 .02
321.04
336 .03
099 .03
NO CURRENT
137 .09
130 .08
159 .05
330 .05
336 .03
004 04
079 .04
053 .03
125 .10
139 .10
098 .04
082 .03
010 .05
347 .03


SEPTEMBER MEAN CURRENT
122° W
121°
120°
119°
118°
117° W
35°
N
34°
33°
32°
C A L I F O R N I A
MEXICO
135 .05
005 .02
006 .05
005 .03
021 .03
014 .03
137 .09
075 .06
068 .03
024 .02
005 .03
014 .04
128 .10
107 .07
102 .05
018 .04
352 .03


OCTOBER MEAN CURRENT
122°W
121°
120°
119°
118°
117°W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
124 .04
043 .02
323 .02
016 .02
143 .08
105 .02
032 .01
016 .02
013 .02
139 .07
108 .04
067 .03
021 .04
023 .02


NOVEMBER MEAN CURRENT
122°W
121°
120°
119°
118°
117°W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
098 .03
021 .03
359 .03
001 .04
034 .02
011 .03
091 .03
048 .04
001 .04
001 .04
097 .04
059 .04
048 .05
027 .04
069 .04
048 .02
114 .06
085 .08
098 .07
336 .02


DECEMBER MEAN CURRENT
122°W
121°
120°
119°
118°
117°W
35°
N
34°
33°
32°
CALIFORNIA
MEXICO
103 .03
115 .02
126 .05
143 .03
319 .02
329 .02
347 .06
029 .02
006 .02
347 .06
004 .03
123 .06
094 .05
097 .03
020 .01
344 .09
350 .05
359 .04
105 .08
110 .05
026 .05
359 .04
112 .07
087 .10
127 .10
142 .08
072 .03
009 .02
332 .04